Casio fx-7700GE
Casio fx-9700GE
Casio CFX-9800G
Casio CFX-9850G

Quadratic Formula Program

This program will display the solutions of a quadratic equation or the words "No Real Solution." To use the program, write the quadratic equation in general form and enter the values of *a, b,* and *c*.

Casio fx-7700GE

Solutions to quadratic equations are also available directly from the Casio calculator's EQUATION MODE.

```
QUADRAT
"AX²+BX+C=0"↵
"A="?→A↵
"B="?→B↵
"C="?→C↵
B²−4AC→D↵
D<0⇒Goto 1↵
(-B+√(D)÷(2A)◢
(-B−√(D)÷(2A)↵
Goto 2↵
Lbl 1↵
"NO REAL SOLUTION"↵
Lbl 2
```

Casio fx-9700GE
Casio CFX-9800G
Casio CFX-9850G

Both real and complex answers are given. Solutions to quadratic equations are also available directly from the Casio calculator's EQUATION MODE.

```
QUADRAT
"AX²+BX+C=0"↵
"A="?→A↵
"B="?→B↵
"C="?→C↵
B²−4AC→D↵
(-B+√(D)÷(2A)◢
(-B−√(D)÷(2A)
```

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists. Solutions to systems of linear equations are also available directly from the Casio calculator's EQUATION MENU.

```
SOLVE
"AX+BY=C"↵
"A=":?→A↵
"B=":?→B↵
"C=":?→C↵
"DX+EY=F"↵
"D=":?→D↵
"E=":?→E↵
"F=":?→F↵
AE−DB=0⇒Goto 1↵
"X=":(CE−BF)÷(AE−DB)◢
"Y=":(AF−CD)÷(AE−DB)↵
Goto 2↵
Lbl 1↵
"NO UNIQUE SOLUTION"↵
Lbl 2
```

Graph Reflection Program

This program will graph a function *f* and its reflection in the line $y = x$. To use this program, enter the function in f1.

Casio fx-7700GE

To use this program, enter the function in f1.

```
REFLECT
"GRAPH -A TO A"↵
"A="?→A↵
Range -A,A,1,-2A÷3,2A÷3,1↵
Graph Y=f1↵
-A→B↵
Lbl 1↵
B→X↵
Plot f1,B↵
B+A÷32→B↵
B≤A⇒Goto1:Graph Y=X
```

Casio fx-9700GE

To use this program, enter a function in f1 and set a viewing rectangle.

```
REFLECT
63Xmin÷127→A↵
63Xmax÷127→B↵
Xscl→C↵
Range , , , A, B, C↵
(Xmax−Xmin)÷126→I↵
Xmax→M↵
Xmin→D↵
Graph Y=f1↵
Lbl 1↵
D→X↵
Plot f1,D↵
D+I→D↵
D≤M⇒Goto 1:Graph Y=X
```

Casio CFX-9800G

To use this program, enter a function in f1 and set a viewing rectangle.

```
REFLECT
63Xmin÷95→A↵
63Xmax÷95→B↵
Xscl→C↵
Range , , , A, B, C↵
(Xmax−Xmin)÷94→I↵
Xmax→M↵
Xmin→D↵
Graph Y=f1↵
Lbl 1↵
D→X↵
Plot f1,D↵
D+I→D↵
D≤M⇒Goto 1:Graph Y=X
```

Casio CFX-9850G

Use the program for the Casio fx-9700GE and replace the line "Range , , , A,B,C,↵" with
"View Window , , , A,B,C,↵."

Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a 2×3 matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the y-intercepts of the lines are between -10 and 10.)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. To use this program, dimension Mat A as a 2×3 matrix. Press [EXE] after each screen display to continue the program.

```
ROWOPS
"ENTER A"↵
"2 BY 3 MATRIX:"↵
"A B C"↵
"D E F"↵
"A="?→A:"B="?→B:
"C="?→C:"D="?→D:
"E="?→E:"F="?→F:↵
[[A,B,C][D,E,F]]→Mat A↵
Cls↵
"ORIGINAL MATRIX:"◢
Mat A◢
Range -10,10,1,-10,10,1↵
Graph Y=B^-1(C−AX)↵
Graph Y=E^-1(F−DX)◢
Cls↵
"OBTAIN LEADING"↵
"1 IN ROW 1"◢
∗Row A^-1,A,1↵
Mat A◢
Graph Y=(A÷B)(C÷A−X)↵
Graph Y=E^-1(F−DX)◢
Cls↵
"OBTAIN 0 BELOW"↵
"LEADING 1 IN"↵
"COLUMN 1"◢
∗Row+ -D,A,1,2↵
Mat A◢
Graph Y=(A÷B)(C÷A−X)↵
GraphY=(E−(BD÷A))^-1(F−(DC÷A))◢
Cls↵
Mat A[2,2]→G↵
G=0⇒Goto 1↵
∗Row G^-1,A,2↵
"OBTAIN LEADING"↵
"1 IN ROW 2"◢
Mat A◢
Graph Y=(A÷B)(C÷A−X)↵
Graph Y=(E−(BD÷A))^-1(F−(DC÷A))◢
Cls↵
"OBTAIN 0 ABOVE"↵
"LEADING 1 IN"↵
"COLUMN 2"◢
Mat A[1,2]→H↵
∗Row+ -H,A,2,1↵
Mat A◢
Mat A[1,3]→J↵
Mat A[2,3]→K↵
Graph Y=K↵
Plot J,-10:Plot J,10:Line◢
"THE POINT OF"↵
"INTERSECTION IS"↵
"X=":J◢
"Y=":K
Goto 3↵
Lbl 1↵
Mat A[2,3]=0⇒Goto 2↵
"INCONSISTENT"↵
"SYSTEM"↵
Goto 3↵
Lbl 2↵
"INFINITELY MANY"↵
"SOLUTIONS"↵
Lbl 3
```

Evaluating an Algebraic Expression Program

This program can be used to evaluate an algebraic expression in one variable at several values of the variable. To use this program, enter an expression in f1.

```
EVALUATE
Lbl 1↵
"X="?→X↵
"F(X)=" : f1◢
Goto 1
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
ADDVECT
Cls↵
"A="?→A↵
"B="?→B↵
"C="?→C↵
"D="?→D↵
Plot 0,0↵
Plot A,B↵
Line↵
Plot 0,0↵
Plot C,D↵
Line↵
A+C→E↵
B+D→F↵
Plot 0,0↵
Plot E,F↵
Line↵
Plot A,B↵
Plot E,F↵
Line↵
Plot C,D↵
Plot E,F↵
Line◢
```

Graphing a Sine Function Program

This program will simultaneously draw a unit circle and the corresponding points on the sine curve. After the circle and sine curve are drawn, you can connect the points on the unit circle with their corresponding points on the sine curve by pressing [EXE]. When starting to write this program, press SHIFT SET UP and select PRM or PARM for the GRAPH TYPE to change to parametric mode.

```
SINESHOW
Rad↵
Range -2.25,π÷2,3,-1.19,1.19,1,0,6.3,.15↵
Graph(X,Y)=(-1.25+cos T,sin T)↵
Graph(X,Y)=(T÷4,sin T)↵
0→N↵
Lbl 1↵
N+1→N↵
Nπ÷6.5→T↵
-1.25+cosT→A↵
sin T→B↵
T÷4→C↵
Plot A,B↵
Plot C,B↵
Line◢
N<12⇒Goto 1↵
Cls
```

Casio CFX-9850G

Use the program for the Casio CFX-9800G and replace the line
"Range -2.25,π÷2,3,-1.19,1.19,1,0,6.3,.15↵"
with the line
"View Window -2.25,π÷2,3,-1.19,1.19,1,0,6.3,.15↵."

Finding the Angle Between Two Vectors Program

This program will sketch two vectors and calculate the measure of the angle between the vectors. Be sure to set an appropriate viewing rectangle.

```
VECANGL
Cls↵
Deg↵
"ENTER (A,B)"↵
"A="?→A↵
"B="?→B↵
"ENTER (C,D)"↵
"C="?→C↵
"D="?→D↵
Plot 0,0↵
Plot A,B↵
Line↵
Plot 0,0↵
Plot C,D↵
Line◢
AC+BD→E↵
√(A²+B²)→U↵
√(C²+D²)→V↵
cos⁻¹(E÷UV)→θ↵
"θ="↵
θ
```